


Задача 2

(Задача 845. Сборник задач по теоретической механике. И.Н. Веселовский. М.: Государственное издательство технико-теоретической литературы, 1955. – 500 с.)

Определить динамические реакции подшипников M и N вала, на который насажен однородный круглый диск радиуса R и массой m так, что его плоскость образует с осью вращения угол α , если угловая скорость постоянна и равна ω , $OM = a$, $ON = b$.

Для решения задачи использовать следующие значения параметров:

$R = 0.3$ м, $m = 1$ кг, $\alpha = 45$ град, $\omega = 1$ рад/с, $a = 0.5$ м, $b = 0.3$ м.




Точный теоретический ответ

Динамические реакции в подшипниках M и N вала одинаковы и определяются по формуле

$$Q = \frac{m \cdot R^2 \cdot \omega^2 \cdot \sin 2\alpha}{8 \cdot (a + b)}$$

Решение задачи в EULER



Система состоит из двух звеньев.

- Станина (body1). Станина является инерциальным звеном и в проекте не отображается.
- Вал (body2).

К валу прикреплены два тела: невесомая ось solid1 и наклонный диск solid2. Нижняя опора вала моделируется шарниром joint1 типа «сферический шарнир». Верхняя опора вала моделируется шарниром joint2 типа «пользовательский шарнир», который имеет все три вращательные степени свободы и одну поступательную вдоль оси вала.

Поступательная степень свободы в верхней опоре позволяет убрать статическую неопределенность восприятия нагрузки вдоль оси вала.

Для задания угловой скорости вращения вала используется объект «условие состояния механизма» condition1.

Динамические реакции опор определяются в горизонтальной плоскости XZ. Для определения реакции нижней опоры, созданы датчики сил F1x и F1z, действующих со

стороны шарнира на опору в направлениях X и Z. Суммарная реакция нижней опоры определяется в датчике F1. Реакция верхней опоры определяется аналогично.

Для определения значений реакций опор достаточно выполнить команду «Расчет сил и ускорений».

Результаты моделирования

При получении точного теоретического решения предполагалось, что диск имеет нулевую толщину. При моделировании в EULER толщина диска solid2 должна быть больше нуля. Поэтому в зависимости от заданной в проекте относительной толщины диска (скаляр h_rel) решение в EULER будет отличаться от теоретического. Эта зависимость представлена в следующей таблице:

Относительная толщина диска h_rel [-]	Относительное отличие от теоретического решения Q_delta_rel [-]
0.1	0.013
0.01	0.000 13
0.001	0.000 001 3
0.0001	0.000 000 014

Текст проекта в EULER

```

scalar R=0.3[m];
scalar m=1[kg];
scalar alfa=45[deg];
scalar w=1[rad/s];
scalar a=0.5[m];
scalar b=0.3[m];
scalar h_rel=0.01;
scalar h=h_rel*2*R;
point point1=point( 0[m], -a, 0[m] );
point point2=point( 0[m], b, 0[m] );
solid solid1=cylinder( point1, point2, 0.01[m] );
point point3=point( 0[m], 0[m], 0[m] );
vector vector1=rotate( projectY, projectZ, 90[deg]-alfa );
point point4=move( point3, vector1, h/2 );
point point5=move( point3, vector1, -h/2 );
solid solid2=cylinder( point4, point5, R, mass = m );
color color1=index( 54 );
body body1=body( color = color1 );
set ground = body1;
color color2=index( 64 );
body body2=body( color = color2 );
body body2 < ( solid1, solid2 );
joint joint1=spherical( body1, body2, point1 );
node node1=nodePoint( point2 );
joint joint2=user( body1, body2, node1, noX:, yesY:, noZ:, yesFiX:, yesFiY:, yesFiZ: );
condition condition1=rotVelocity( body1, projectY, body2, w );
sensor F1x=jointForce( force:, body1, point1, projectX, joint1 );
sensor F1z=jointForce( force:, body1, point1, projectZ, joint1 );
sensor F1=sqrt(F1x*F1x+F1z*F1z);
sensor F2x=jointForce( force:, body1, point2, projectX, joint2 );
sensor F2z=jointForce( force:, body1, point2, projectZ, joint2 );
sensor F2=sqrt(F2x*F2x+F2z*F2z);
sensor Q_theoretical=m*R*R*(w/1[rad])*(w/1[rad])*sin(2*alfa)/(8*(a+b));
sensor Q_delta_rel=abs((F1-Q_theoretical)/Q_theoretical);

```

Λ//

Λ Единицы измерения;

set units = SI;